

```
/* Five_yr_avg_HPUE_for presentation.sas
```

```
Purpose: Program to bring in the individual HPUE data from each site, species,  
and year and to calculate the 5 yr average by week for each
```

```
Written by: Diana Tersteeg
```

```
Data:
```

```
Inputs: CAT.site_year_hpue
```

```
Outputs: S:\creel\Catch_Rates\Multi Year\TEST_&SITE._HPUE_&YEAR1._&YEAR4._&sysdate..xml
```

```
Notes:
```

```
Modifications: Art provided the proc format code and I simplified the first data step to pull  
in all the data.
```

```
*/
```

```
Libname CAT 'S:\creel\Catch_Rates\_SAS datasets';
```

```
OPTIONS MPRINT MPRINTNEST symbolgen ;
```

```
ods _all_ close;
```

```
/* Creating multiple Year lines so can easily define which 5 years you want to use in the macro  
call Could modify to use fyear and lyear and have macro define year1-year5 later */
```

```
%MACRO CAT_GO(INLIB=CAT,  
SITE=,  
NUMYR=4,  
YEAR1=2011,  
YEAR2=2012,  
YEAR3=2013,  
YEAR4=2014,  
YEAR5=,  
REDCT=Y,  
SAVELIB=CAT);
```

```
proc format;
```

```
value $sitecode
```

```
"KETCHIKAN" = 'A_KTN_';
```

```
"POW" = 'B_POW_';
```

```
"PETERSBURG" = 'C_PTG_';
```

```
"WRANGELL" = 'C_WRG_';
```

```
"SITKA" = 'D_SIT_';
```

```
"JUNEAU" = 'E_JNU_';
```

```
"GUSTAVUS" = 'G_GUS_';
```

```
"ELFIN COVE" = 'G_elf_';
```

```
"YAKUTAT" = 'H_YAK_';
```

```
other = 'UNK_';
```

```
run;
```

```
/* If you place an equal sign between the ampersand and the macro variable name of a direct macro  
variable reference, the macro variables name displays in the log along with the macro variables  
value. Same as %put Var=&var; in 9.2 and prior */
```

```

%let n = %sysfunc(putc(&site,$sitecode.));
%put &n;

%let P=&n&YEAR1._&YEAR4.;

/* Re-direct the log output to the output log file if REDCT macro var in the main call is set to
Y */

%IF &REDCT EQ Y %THEN %DO;

    PROC PRINTTO NEW LOG="S:\creel\Catch_Rates\Multi Year\log files\&P._HPUE_1.LOG";
    RUN;
%END;
%ELSE;

/* START OF MAIN CODE */

/* Grabbing each of the defined years datasets for defined port and putting them in one file
The next macro will pull out each species for each call of the macro NAME
*/

DATA &N.ALL;
    set
        %do I=&YEAR1. %to &YEAR4. %by 1;
            &INLIB..&N.&I._hpue (keep=site year week tking tchoho tpink tchum thalibut trockfish)
        %end;
    ;
run;

/* Because we want a different xls file for each port, this ods stmt needs to be inside the
CAT_GO macro but outside NAME macro. The NAME macro does all the sheet tabs within the
spreadsheet
*/

ods tagsets.ExcelXP file=
    "S:\creel\Catch_Rates\Multi Year\&SITE._HPUE_&YEAR1._&YEAR4._&sysdate._1.xml"
    style=meadow options(embedded_titles='yes' embedded_footnotes='yes');

%Macro Name(spec);

    Data &N&spec;
        set &N.All;

    keep site year week t&spec;
    run;

    proc sort data=&N&spec;
    by SITE WEEK;
    run;

/* Proc transpose to give the data in output desired */

proc transpose data=&N&spec out=&N&spec._out(drop=_NAME_) prefix=T&spec._;
    by SITE week;
    id YEAR;
run;

```

```
/* Use Proc Means with MEAN as it is the Mean (Average) of the analysis variable's non-missing
values. Therefore, if a survey was not done because the season had not started, then that
variable will not be included because it SHOULD be set as missing vs 0 */
```

```
proc means data=&N&spec noprint;
  by SITE WEEK;
  var T&spec;
  title "&NUMYR. year average (&year1 - &Year4) of HPUE for &SITE for &spec";
  output out=&N&spec._avg (drop= _type_)
    mean=T&spec._MEAN;
run;
title;
```

```
/* Merge 5 years of data with mean data
   right now these are work files - can make perm */
```

```
data &P._&spec;
  merge &N&spec._out
        &N&spec._avg;
  by SITE WEEK;
run;
```

```
/* For visual effect - replace null with M or _ (.) or any single letter
   used .N - for No survey this week */
```

```
data &P._&spec._d;
  set &P._&spec;
  If t&spec._&year1=. then t&spec._&year1=.N;
  If t&spec._&year2=. then t&spec._&year2=.N;
  If t&spec._&year3=. then t&spec._&year3=.N;
  If t&spec._&year4=. then t&spec._&year4=.N;
  * If t&spec._&year5=. then t&spec._&year5=.N;
run;
```

```
/* Calculate hours per fish = reciprocal of HPUE*/
```

```
data &P._&spec._h;
  set &P._&spec._d;

  hr_t&spec._&year1=1/t&spec._&year1;
  hr_t&spec._&year2=1/t&spec._&year2;
  hr_t&spec._&year3=1/t&spec._&year3;
  hr_t&spec._&year4=1/t&spec._&year4;
  * hr_t&spec._&year5=1/t&spec._&year5;
  hr_t&spec._MEAN=1/t&spec._MEAN;
run;
```

```
/* Output all current species information to one tab/sheet of spreadsheet */
```

```
ods tagsets.ExcelXP options(sheet_name="&SPEC");
options Missing='*';
proc print data=&P._&spec._h noobs;
  VAR SITE WEEK T&spec._&year1 T&spec._&YEAR2 T&spec._&YEAR3 T&spec._&YEAR4 /*T&spec._&year5*/
  T&spec._MEAN hr_t&spec._&year1 hr_t&spec._&year2 hr_t&spec._&year3 hr_t&spec._&year4
  /*hr_t&spec._&year5*/ hr_t&spec._MEAN;
run;
```

```

format T&spec._&year1 T&spec._&YEAR2 T&spec._&YEAR3 T&spec._&YEAR4 /*T&spec._&year5*/
T&spec._MEAN 8.4 hr_t&spec._&year1 hr_t&spec._&year2 hr_t&spec._&year3 hr_t&spec._&year4
/*hr_t&spec._&year5*/ hr_t&spec._MEAN 5.0;
title "HPUE for &YEAR1 to &YEAR4 and Mean &NUMYR. year HPUE in &SITE for &SPEC";
footnote1 "Tspecies=Average number of the targeted species harvested per rod hour: 0 = no
harvest, N = Survey not done this week.";
footnote2 "hr_Tspecies=Average number of hours it takes to harvest the targeted species: * = No
survey or 0 harvest.";
footnote3 "2014 Halibut HPUE calculations include GAF Halibut harvest";
footnote4 "Run on &sysdate9 at &systemtime by DTersteeg.";
run;

/* close inner macro */
%mend NAME;

/* inner macro calls */
%NAME (KING)
%NAME (COHO)
%NAME (PINK)
%NAME (CHUM)
%NAME (HALIBUT)
%NAME (ROCKFISH)

/* close excel sheet for current port once all calls for macro NAME have run */
ods tagsets.ExcelXP close;

/* clear all titles and footnotes from prior port */
title;
footnote;

/* output log to file if Y in CAT_GO
if not then %PUT _user_ will print out user macro variables to system log*/
%IF "&REDCT" EQ "Y" %THEN %DO;

PROC PRINTTO;
RUN;

%END;
%ELSE;
%PUT _user_;

/* close outer macro */
%mend CAT_GO;

/* outer macro calls */
%CAT_GO (SITE=KETCHIKAN);
%CAT_GO (SITE=POW);
%CAT_GO (SITE=PETERSBURG);
%CAT_GO (SITE=WRANGELL);
%CAT_GO (SITE=SITKA);
%CAT_GO (SITE=JUNEAU);
%CAT_GO (SITE=ELFIN COVE);
%CAT_GO (SITE=GUSTAVUS);
%CAT_GO (SITE=YAKUTAT);

```